Calculating the Carstairs index for the 2001Census

The Carstairs index uses four variables derived from the census:

- Proportion male unemployment
- Proportion overcrowded households
- Proportion no car/vans ownership
- Proportion low social class

These variables have been constructed from the following 2001 Census tables

Male Unemployed:
CS021: Males Unemployed 16-74 / Economically Active Males 16-74
CS0210049 / CS0210013

Overcrowded Households:
CS052: (Over 1 and up to 1.5 persons per room + Over 1.5 persons per room) / All Households
(CS0520013 + CS0520017) / CS0520001

No Car
KS017: No Cars or vans in household / All households
KS0170002 / KS0170001

Low Social Class
UV31: (L11.2+L12.2+L12.4+L12.5+L12.7+L13.1+L13.2+L13.4+L13.5) / All persons
(UV0310031 + UV0310034 + UV0310036 + UV0310037 + UV0310039 + UV0310041 + UV0310042 + UV0310044 + UV0310045 + UV0310046) / UV0310001

The low social class variable is an approximation to the previous Social Class census variable, because the ONS changed the method of occupational social class classification. A user guide is available at http://www.statistics.gov.uk/methods_quality/ns_sec/continuity.asp .The variables extracted from table UV31 approximate to Social Class IV and Social Class V. N.B. Some warnings have been given about the quality of the tables with NSSEC, especially about 'never worked' cell 0047, but this variable is not used here.

For wards with over 100 households:

```
compute unemp = m_unemp / m16_74.
compute overcrow = (ppr1 +  ppr1_5) / all_hh2.
compute nocar = no_car / all_hh1.
compute lowclass = lowjob / p16_74.

RECODE
  unemp overcrow nocar lowclass unemp (SYSMIS=0).
EXECUTE .

DESCRIPTIVES
  VARIABLES= unemp  overcrow nocar lowclass /SAVE.
execute.

compute carstair = (zunemp + zovercro + znocar + zlowclas).

RANK
  VARIABLES = carstair
  /NTILES(5)
  /PRINT = NO
  /TIES = MEAN.

execute.
```

Calculating the Townsend index for the 2001Census

The Townsend index uses four variables derived from the census:

- Percentage unemployment
- Percentage overcrowded households
- Percentage no car/vans ownership
- Percentage non-home owners

These variables have been constructed from the following 2001 Census tables

Unemployed:
CS021: Unemployed 16-74 / Economically Active 16-74 * 100
CS0210046 / CS0210010 * 100

Overcrowded Households:
CS052: (Over 1 and up to 1.5 persons per room + Over 1.5 persons per room) / All Households * 100
(CS0520013 + CS0520017) / CS0520001 * 100

No Car
KS017: No Cars or vans in household / All households * 100
KS0170002 / KS0170001 * 100

Non-home ownership
KS018: Renting home / All households * 100
(KS0180005 + KS0180006 + KS0180007 + KS0180008) / KS0180001 * 100

For wards with over 100 households:

```
compute unemp = (unempl / econact) * 100.
compute overcrow = ((ppr1 +  ppr1_5) / all_hh2) * 100.
compute nocar = (no_car / all_hh1) * 100.
compute renting = (renters / all_hh) * 100.

RECODE
  unemp overcrow nocar renting(SYSMIS=0).
EXECUTE .

compute unemp = ln(unemp + 1).
compute overcrow = ln(overcrow + 1).

DESCRIPTIVES
  VARIABLES= unemp overcrow nocar renting /SAVE.
execute.

compute townsnd = (zunemp + zovercro + znocar + zrenting).

RANK
  VARIABLES = townsnd
  /NTILES(5)
  /PRINT = NO
  /TIES = MEAN .

execute.
```

Paul Norman
Cathie Marsh Centre for Census and Survey Research (CCSR)
paul.norman@manchester.ac.uk